ice felix®

hc "90



întreprinderea de calculatoare electronice

78009, 2 G.Constantinescu st., Bucharest 2,
ROMANIA, tel. 886030, telex 11626 - felix r.

INSTRUCTIUNI DE INSTALARE

FELIX HC90

ICE FELIX

INTRODUCERE

Acest manual este facut cu intentia de a va ghida primi pasi in utilizarea calculatorului FELIX HC90.
Inaintea orcarei instalari cititi acest manual. El cuprinde:

- Punerea sub tensiune
- Tastatura
- Salvarea programelor pe caseta
- Ce urmeaza...

INAINTE DE PORNIRE:

* *Verificati configuratia standard :*

- unitatea centrala cuprinzind si tastatura
- sursa de alimentare (alimentatorul)
- cablu pentru televizor
- cablu pentru casetofon
- prezentul manual
- caseta de demonstratii

PUNEREA SUB TENSIUNE

Calculatorul HC90 se instaleaza usor si rapid. Urmati pentru aceasta secventa:

1. Introduceti fisa alimentatorului in mufa aflata pe partea laterala dreapta a calculatorului.

2. Introduceti stecherul alimentatorului intr-o priza de curent alternativ 220 V/50 Hz.

In acest moment HC90 functioneaza. Daca apasati tastele veti auzi bipuri sonore. Daca nu le auziti apasati butonul RESET si incercati din nou. In momentul in care apasati tastele HC90 primeste comenzile dumneavoastra. Ca sa puteti dialoga aveti nevoie de un dispozitiv de afisare. Cel mai simplu este un televizor alb/negru sau color.

3. Conectati cablul de televizor: Introduceti un capat al cablului in mufa de antena a televizorului si celalalt capat in mufa de pe calculator notata TV.

4. Aprindeti televizorul, acordati-l pentru canalul 8 pina cind obtineti o imagine neta.

HC90 este gata de lucru. Pentru o mica demonstratie a posibilitatilor sale folositi caseta de demonstratie:

5. Conectati cablul de casetofon (DIN cu cinci picioruse intre calculator si casetofon.

6. Introduceti caseta in casetofon pozitionata la inceput.

7. Introduceti comanda LOAD "" si apasati ENTER. Pentru aceasta apasati apasati tasta J (pe ecran apare scis LOAD) si apoi tinind apasata tasta SYMBOL SHIFT apasati de doua ori tasta P (pe ecran apar de doua ori ghilimele). Apasati tasta CR.

8. Porniti casetofonul. Pe marginile ecranului apar dungi colorate. Se incarcaa programul. Urmariti si executati mesajele de pe ecran.

Dupa ce v-ati familiarzat putin cu tastatura si caseta folosind programul de demonstratie cititi mai departe manualul:

TASTATURA

Ati vazut cum se introduce comanda de "incarcare program". V-ati obisnuit putin cu tastatura. Desi are numai 40 de taste claviatura este la inceput greu de utilizat fiecare tasta avind pe ea trei patru sau chiar cinci simboluri.

Sa incercam sa tastam ceva.

Mai intii apasati butonul RESET pe ecran apare mesajul FELIX HC90. Sinteti pregatiti pentru a dialoga cu calculatorul. Cuvintile pe care le intelege HC 90 sint specifice limbajului BASIC. Apasind tastele calculatorului puteti

introduce litere, cifre separate sau cuvinte BASIC. Pentru a intelege modul de lucru al tastaturii sa privim cum sint inscriptionate tastele. Fiecare tasta are in centru un simbol (litera sau cifra). Daca ne limitam la aceste simboluri constatam ca tastele sint asezate ca la o masina de scris obisnuita.

Ln afara de simbolul central tasta mai cuprinde doua simboluri asezate in partea dreapta si doua simboluri asezate in partea stinga. Cind apasati o tasta HC90 va trebui sa-si aleaga unul din simbolurile de pe ea.

Pentru a-i indica ce inscriptie sa-si aleaga avem la dispozitie doua taste speciale numite CAPS SHIFT si SYMBOL SHIFT. Ln afara de acestea mai conteaza si modul de lucru care este afisat de un cursor clipitor. Modul de lucru poate fi:

-K " keyword" cuvint cheie, HC90 foloseste atunci un cunint BASIC de pe tastatura;

-L mod in care se foloseste simbolul central de pe tasta

-E mod extins utilizeaza inscriptiile de pe partea din stinga tastei;

-G modul grafic, in care se pot utiliza simbolurile semigrafice de pe primul rind al tastaturii (tastele numerice 0-9)

Trecerea dintr-un mod in altul se face in general automat pe masura apasarii tastelor HC90 ne ajuta sa respectam structura unei linii de comanda sau program BASIC.

Totusi trecerea din modul normal in modul extins si invers se face prin apaasare simultana a celor doua taste SHIFT (SYMBOL SHIFT si CAPS SHIFT).

CE URMEAZE MAI DEPARTE...

Daca vreti sa utilizati calculatorul ca "beneficiar" cu programe "gata facute" nu mai aveti practic multe de invatat. Un scurt ghid de BASIC pentru HC 90 ne este suficient. De obicei insa in practica doriti sa programati activitati pentru care ori nu exista ori nu aveti acele programe. Singura solutie este sa invatati sa va faceti singuri aceste programe. Pentru inceput veti consulta manuale BASIC pentru HC90 iar pentru aplicatii mai evoluate veti folosi limbajul cod masina.

ELEMENTE DE PROGRAMARE SI EDITARE

Utilizarea tastaturii

Tastatua calculatorului HC-90 este similara cu o masina de scris. Literele si cifrele sint in aceleasi pozitii cu exceptia literelor Q, Z, M. Tastatura cuprinde simboluri simple (litere, numere, etc) si compuse (cuvinte cheie, nume de functii etc) care sint introduse printr-o singura apasare si nu prin tastare caracter cu caracter. Pentru a obtine toate functiile si comenzile unele taste au pina la sase semnificatii diferite, selectionabile prin actionarea tastei corespunzatoare simultan cu una din tastele CAPS SHIFT sau SYMBOL SHIFT si in functie de modul de lucru al calculatorului.



Fig. 2.1

HC-90 are cinci moduri de lucru :

1. Modul K (keyword-cuvint cheie) apare atunci cind se selecta o comanda sau o linie de program (altceva decit date de intrare). Aceasta se intimpla la inceput de linie, imediat dupa un THEN, sau dupa caracterul ":" (ce separa instructiuni aflate pe aceeasi linie). In modul K, daca nu sint utilizate shift-urile, tasta urmatoare va fi interpretata ca un cuvint cheie (pozitia 3 din figura pentru taste alfanumerice) sau ca o cifra (pozitia 1 din figura pentru taste numerice).

2. Modul L (letters-litere) alterneaza cu modul K. Simbolul principal scris pe o tasta apare pe ecran prin simpla actionare a acesteia. In cazul unei litere, ea va apare ca litera mica. Atit in modul L cit si in modul K actionarea simultana a lui SYMBOL SHIFT si a unei taste numerice va fi interpretata drept caracterul din pozitia 3 iar cu o tasta literala produce caracterul din pozitia 2 (vezi fig. 2.1). CAPS SHIFT cu o tasta numerica se va interpreta ca functia de control din pozitia. Actionarea unei taste literale simultan cu CAPS SHIFT in modul de lucru K nu are nici un efect, iar in modul L produce conversia literelor mici in litere mari.

3. Modul C (capitals-majuscule) este o varianta a modului L, in care srierea se face cu majuscule. CAPS LOCK determina trecerea din modul L in modul C si invers.

4. Modul E (extended-extins) este utilizat pentru a obtine caractere noi, in special alte comenzi. Aceast mod se obtine prin actionarea simultana a ambelor shift-uri si se anuleaza automat dupa prima tastare. In acest mod apasarea unei taste literale genereaza caracterul sau simbolul de pe pozitia 4 daca e apasata singura si caracterul sau simbolul de pe pozitia 5 daca e apasata impreuna cu una din tastele SHIFT. Apasarea unei taste numerice genereaza o comanda daca e apasata impreuna cu SYMBOL SHIFT si o secventa de control a culorii daca e apasata singura.

5. Modul G (grafics-grafic) se obtine prin actionarea tastelor CAPS SHIFT si 9. Anularea acestui mod de lucru se face actionind din nou tastele CAPS SHIFT si 9 sau numai tasta 9. O tasta numerica va da un mozaic grafic predefinit (in afara tastelor 0 si 9) si orice tasta literala in afara de V,W,X,Y si Z va genera un simbol grafic definit de utilizator.

Daca o tasta este apasata mai mult de 2-3 secunde aceasta va incepe sa se repete.

Tastele apasate apar in partea de jos a ecranului, fiecare caracter fiind inserat pe locul cursorului. Cursorul poate fi mutat la stinga cu <-- (CAPS SHIFT si 5) sau la dreapta cu --> (CAPS SHIFT si 8). Caracterul din stinga cursorului poate fi sters cu DELETE (CAPS SHIFT si 0).

La inscrierea simbolurilor pe tastatura au fost folosite urmatoarele prescurtari:

RAND	in loc de	RANDOMIZE
BRGT	in loc de	BRIGHT
INV	in loc de	INVERSE
CR	in loc de	ENTER
CS	in loc de	CAPS SHIFT
SS	in loc de	SYMBOL SHIFT
SCR$	in loc de	SCREEN$
CONT	in loc de	CONTINUE

2.2 Modul de afisare

Ecranul de afisare are 24 de linii, fiecare cu 32 de caractere.

Ecranul are doua parti. Partea de sus de 22 linii e folosita pentru listarea instructiunilor sau a rezultatelor programului. Cind aceasta parte este plina calculatorul face "scroll". Pentru a putea vedea toate liniile, calculatorul se opreste si apare mesajul "scroll?". Apasarea tastelor N,SPACE sau STOP va intrerupe programul si va afisa mesajul

D BREAK - CONT repeats
Orice alta tasta determina calculatorul sa faca scroll. Partea de jos a ecranului este folosita pentru comenzi de intrare, linii de program, tiparirea datelor de intrare cit si pentru mesaje.

Programe, linii de program si editarea programelor utilizind EDIT si sagetile, RUN, LIST, GO TO, CONTINUE, INPUT, NEW, REM, PRINT, STOP in INPUT data, BREAK

Limbajul BASIC admite doua tipuri de instructiuni: numerotate si nenumerotate. Instructiunile nenumerotate sint executate imediat dupa apasarea tastei CR. Instructiunile numerotate sint stocate ca linii de program. Numerele de linie trebuie sa fie intregi, intre 1 si 9999. Listarea si executia unui program se fac prin ordonarea programului dupa numerele de linie. De aceea este indicat ca la scrierea unui program sa se pastreze spatii intre numerele a doua linii consecutive, dind astfel posibilitatea inserarii cu usurinta de linii noi. O linie de program poate contine una sau mai multe instructiuni separarea instructiunilor dintr-o linie se face cu caracterul ":".

Cursorul > indica linia curenta asupra careia se pot face modificari sau dupa care se pot insera alte linii. De obicei, cursorul se afla pe ultima linie introdusa, dar pozitia lui poate fi deplasata in sus sau in jos prin apasarea simultana a tastei CAPS SHIFT si a sagetilor.

In continuare vor fi prezentate exemple de programe in care sint trecute in revista citeva instructiuni BASIC, punindu-se accentul pe facilitatile de editare ale sistemului.

Exemplul 1. Sa se tipareasca suma a doua numere .

Dupa ce se vor introduce liniile (in ordinea mentionata):

```
20 PRINT a
10 LET a=10
```

se constata ca programul se tipareste pe ecran in permanenta ordonat dupa numarul de linie.

Pina acum s-a introdus primul numar. Pentru a-l introduce pe al doilea, se scrie linia:

```
15 LET b=15
```

Pentru tiparirea sumei, este necesar ca linia 20 sa aiba forma

```
20 PRINT a+b
```

S-ar putea rescrie linia, dar este mai usor sa se faca uz de facilitatile EDIT. Pentru aceasta se coboara cursorul de la linia 15 la linia 20, actionind tasta !. In continuare se actioneaza tasta EDIT; in partea de jos a ecranului va apare o

copie a liniei curente (in exemplul prezentat, linia 20). Se actioneaza tasta --> pina cind cursorul L se deplaseaza la sfirsitul liniei si apoi se introduc +b (fara CR).
Ultima linie a ecranului va arata acum astfel:

```
20 PRINT a+b
```

Cu CR, vechea linie 20 va fi inlocuita cu cea noua.

Se executa acest program utilizind RUN si CR; ca urmare pe ecran va apare afisat rezultatul operatiei a+b. Apasind din nou RUN si CR programul este executat identic. Dupa terminarea executiei programului ramine inregistrata in memorie ultima valoare a fiecarei variabile din program. Ele pot fi vizualizate printr-o instructiune PRINT neetichetata. Aceasta operatie este utila la depanarea programului.

Pentru a sterge ultima linie a ecranului se utilizeaza EDIT. Se introduce o succesiune de caractere (fara CR) care vor fi sterse folosind una dintre metodele:

1. actionarea tastei DELETE pina cind linia este stearsa in intregime.
2. actionarea tastei EDIT; pe ultima linie a ecranului apare o copie a liniei curente. Cu CR acum, linia curenta ramine nemodificata, iar ultima linie a ecranului este stearsa.

Presupunem ca se introduce din greseala linia:

```
12 LET b=8
```

Ea va putea fi stearsa scriind:

```
12                    (cu CR.desigur).
```

Se observa ca a disparut cursorul programului. Daca se actioneaza ! , cursorul va apare la linia 10, in timp ce daca se actioneaza ! va apare la linia 15. Se scrie:

```
12                    (si CR)
```

Din nou cursorul programului va fi ascuns intre liniile 10 si 15. Actionind acum EDIT, linia 15 va apare in zona de editare. Cind cursorul programului este ascuns, intre doua linii, EDIT aduce in josul ecranului linia care are numarul de linie imediat urmator.

Se scrie acum:

```
30                    (si CR)
```

De aceasta data cursorul este ascuns dupa sfirsitul programului.

Cu comanda

```
LIST 15
```

pe ecran se obtine:

```
15>LET b=15
20 PRINT a+b
```

Instructiunea LIST 15 produce listarea incepind cu linia 15 si pune cursorul programului la linia 15. Pentru un program foarte lung, LIST va fi o metoda mai utilizata de mutare a cursorului decit | , | .

Aceasta ilustreaza o alta utilitate a numerelor de linie: ele actioneaza ca nume ale liniilor de program astfel incit se pot face referiri la ele in acelasi mod in care se fac referiri la numele de variabile. LIST (neurmat de un numar) determina listarea de la inceputul programului.

O alta comanda este NEW . Efectul ei consta in stergerea programelor si variabilelor din memoria calculatorului.

Exemplul 2. Sa se scrie un program care transforma temperatura din grade Fahrenheit in grade Celsius.

```
10 REM conversia temperaturii
20 PRINT "grade F","grade C"
30 PRINT
40 INPUT "introduceti gradele F.  ",f
50 PRINT f,(f-32)*5/9
60 GO TO 40
```

Este necesar sa fie introdusa pe rind fiecare litera pentru a obtine "conversia temperaturii" in linia 10. In linia 60 se obtine GO TO actionind tasta G (desi contine spatiu, GO TO constituie un singur cuvint cheie).

Rulind programul, se va vedea pe ecran capul de tabel tiparit de linia 20. Linia 10 este ignorata de calculator, instructiunea REM introducind un comentariu in textul sursa. Comanda INPUT din linia 40 asteapta sa fie introdusa o valoare pentru variabila F; se introduce un numar si se actioneaza apoi CR. Calculatorul afiseaza rezultatul si nu se opreste din rulare, ci asteapta alt numar (datorita saltului din linia 60). Programul se poate opri prin actionarea tastei STOP in momentul in care pe ecran apare scris:

Introduceti gradele F.

Calculatorul intoarce mesajul

H STOP in INPUT 40:1

care precizeaza de ce si unde s-a oprit din rulare (in prima instructiune din linia 40).

Pentru a continua programul se introduce CONTINUE si calculatorul va astepta alt numar. CONTINUE determina rularea programului de la linia de la care se oprise executia (linia 40).

Se rescrie linia 60 sub forma

60 GO TO 31

In executie aceasta varianta se comporta identic cu varianta precedenta. Daca numarul liniei intr-o comanda GO TO se refera la o linie inexistenta, atunci se sare la linia imediat urmatoare numarului dat. Acest lucru este valabil si pentru comanda RUN (de fapt RUN are acelasi efect cu RUN 0).

Daca tiparim numere pina cind se umple ecranul calculatorul va muta intreaga parte de sus a ecranului cu o linie pentru a face loc, pierzind astfel capul de tabel. Cind am terminat de tiparit, programul se poate opri cu STOP urmat de CR. Lista de instructiuni a programului se poate afisa dupa intrerupere apasind CR.

Se analizeaza instructiunea PRINT din linia 50. Virgula utilizata aici determina inceperea tiparirii fie in marginea din stinga, fie in mijlocul ecranului, in functie de ce urmeaza dupa virgula. In acest caz tiparirea temperaturii in grade Celsius are loc in mijlocul liniei.

Caracterul punct si virgula ";", determina tiparirea sirului urmator imediat dupa sirul precedent. Se poate vedea aceasta daca in linia 50 e inlocuit caracterul "," cu ";".

Alt semn de punctuatie ce poate fi utilizat in comenzi PRINT este apostroful "'". El determina saltul cursorului la inceputul liniei urmatoare si continuarea tiparirii din acel punct, ca si cum elementele despartite prin "'" ar fi fost sub incidenta unor comenzi PRINT succesive. Pentru ca instructiunea PRINT sa nu determine saltul la linia urmatoare este necesar ca PRINT-ul precedent sa se termine cu "," sau cu ";". Pentru exemplificare sa se substituie linia 50 pe rind cu liniile:

```
50 PRINT f,
50 PRINT f;
50 PRINT f
50 PRINT f'
```

Se constata ca varianta cu "," imparte totul in doua coloane, cea cu ";" scrie totul compact, cea fara semn de punctuatie si cea cu "'" scriu un numar pe o linie.

In memorie pot exista simultan mai multe programe cu conditia ca numerele de linie sa fie in intervale disjuncte.

Exemplul 3.

```
100 INPUT n$
110 PRINT "Salut ";n$ " !"
120 GO TO 100
```

Acesta este un program care poate coexista in memorie cu programul din exemplul 2 intrucit unul are numerele de linie in intervalul 0...60 iar celalalt in 100...120. Pentru lansarea in executie a programului din exemplul 3 se da comanda RUN 100. Executia unei comenzi RUN determina stergerea ecranului si a tuturor variabilelor, dupa aceasta executind sirul instructiunilor programului. Daca nu se doreste initilizarea variabilelor si stergerea ecranului se poate utiliza comanda GO TO 100.

La executia programului din exemplul 3 se observa ca pe ecran apare "L" care indica faptul ca se doreste citirea unui sir de caractere. Sistemul admite ca o instructiune INPUT sa se comporte similar cu o instructiune de atribuire, dar numai pentru cazul citirii de variabile de tip sir de caractere. Pentru aceasta se sterg ghilimelele (utilizind <-- si DELETE) si se introduce numele unei variabile de acelasi tip. Introducerea unui nume de variabila determina cautarea valorii acelei variabile ce trebuia citita de la tastatura.

De exemplu daca la executia programului din exemplul 3 la prima solicitare de sir de caractere se introduce "ANA" , valoarea variabilei n$ va deveni n$="ANA" la urmatoarea citire se introduce "MARIA", n$ devine n$="MARIA" . La executia urmatoarei instructiuni INPUT se va introduce n$; in acest caz se cauta

valoarea vechii variabile n$ si i se asociaza variabilei n$. Deci comanda se comporta similar cu LET n$=n$ .Valoarea lui n$ in urma acestei instructiuni va fi n$="MARIA" , deci instructiunea PRINT din linia 110 va tipari:

Salut MARIA !

Uneori din greseala se scrie un program ce ruleaza la infinit, cum este urmatorul:

```
200 GO TO 200
RUN 200
```

Pentru oprirea executiei se actioneaza BREAK (CAPS SHIFT si SPACE) si calculatorul raspunde cu mesajul

L BREAK into program, 200:1

La sfirsitul fiecarei instructiuni programul verifica daca aceste taste sint actionate; daca da, este oprita rularea.

Tasta BREAK poate fi utilizata de asemenea cind sint conectate casetofonul sau imprimanta, in cazul cind calculatorul asteapta ca aceste periferice sa efectueze o comanda. Mesajul produs in acest caz este diferit:

D BREAK - CONT repeats.

Comanda CONTINUE in cazul lucrului cu casetofonul sau imprimanta repeta instructiunea unde programul a fost oprit.

Listingurile automate sint acelea care nu rezulta in urma unei comenzi LIST, ci au loc dupa introducerea unei linii noi. De retinut este faptul ca linia curenta (cea cu >) apare intotdeauna pe ecran si in mod normal in pozitie centrala. Calculatorul memoreaza numarul liniei curente si de asemenea, al primei linii din partea de sus a ecranului.

Cind incearca sa listeze, primul lucru pe care-l face este sa compare prima linie de pe ecran cu linia curenta. Daca prima linie de pe ecran este mai mare decit linia curenta, atunci cursorul va apare pe prima linie a ecranului. Altfel listarea consta in tiparirea pe ecran in mod defilare a programului cuprins intre prima linie si linia curenta.

Oricum, mai intii se efectueaza un calcul aproximativ pentru a vedea cit timp ia listarea si daca acesta este prea lung, linia din virf se muta mai jos pentru a fi mai aproape de linia curenta. Acum, avind stabilita linia din virf, listarea poate incepe. Daca linia curenta a fost listata, listarea se opreste cind s-a ajuns la sfirsitul programului sau la partea de jos a ecranului.

LIMBAJUL BASIC

Variabile si expresii aritmetice

Cuprins: Sume de variabile, expresii, notatii
Operatii: +,-,*,/

Versiunea BASIC a calculatorului HC 90 admite pentru variabilele numerice nume formate din oricite caractere (litere sau cifre), care incep cu o litera. Printre caractere poate fi si blancul, care este insa ignorat. Prezenta lui face variabila mai usor de citit. Sistemul face filtrarea literelor mari, astfel incit, atit litera mare cit si litera mica corespunzatoare sint interpretate la fel. Nu este indicata folosirea numelor foarte lungi deoarece sint greu de manipulat.

Variabilele speciale sint:

1. Variabilele folosite in instructiunile FOR, care trebuie sa fie reprezentate printr-o singura litera.
2. Variabilele de tip sir de caractere, al caror nume este format dintr-o litera urmata de "$".

Expresiile numerice pot fi reprezentate si printr-un numar zecimal urmat de un exponent.

Exemplul 1. Sa se tipareasca numerele:

```
PRINT 2.3e0
PRINT 2.34e1
```

si asa mai departe pina la

```
PRINT 2.34e15
```

Se observa ca dupa un timp calculatorul incepe sa foloseasca scrirea cu exponent deoarece nu se pot utiliza mai mult de 14 caractere consecutive pentru scrierea unui numar.

Se poate tipari in mod similar:

```
PRINT 2.34e-1
PRINT 2.34e-2
```

si asa mai departe. Comanda PRINT afiseaza numai 8 cifre semnificative.

Exemplul 2.

```
PRINT 4294967295,4294967295-429e7
```

Acest exemplu demonstreaza ca toate cifrele numarului 4294967295 sint memorate, desi nu toate pot fi tiparite pe ecran.

HC 90 utilizeaza scrierea numerelor in virgula mobila.

Numerele sint reprezentate cu precizie de aproximativ noua cifre si jumatate. Cel mai mare intreg ce poate fi reprezentat cu precizie in memorie este 2e32-1=4294967295.

Exemplu:

```
PRINT 1e10+1-1e10,1e10-1e10+1
```

Rezultatele afisate vor fi

0 1

deoarece 1e10+1 si 1e10 au aceeasi reprezentare interna.

Operatiile aritmetice executate de calculator sint inmultirea, impartirea, adunarea si scaderea. Operatiile de inmultire "*" si impartire "/" au prioritate egala. De aceea, o expresie ce contine numai inmultiri si impartiri se executa de la stinga la dreapta. Adunarea si scaderea au de asemenea, prioritate egala dar mai mica decit a inmultirii si a impartirii .

Pentru a modifica ordinea de executie a operatiilor se folosesc parantezele.

Siruri de caractere

Cuprins: Operatii cu siruri de caractere

Sirurile de caractere sint reprezentate prin secvente de caractere ASCII, incadrate intre ghilimele ("). Daca se doreste tiparirea in text a caracterului ghilimele, el trebuie sa fie dublat. Un sir de caractere poate fi atribuit ca valoare unei variabile sir sau poate fi tiparit cu o comanda PRINT.

Fiind dat un sir, un subsir al lui consta in citeva caractere consecutive continute in el, luate in secventa. De exemplu "string" este un subsir al lui "bigger string", insa "b string" nu este. Manipularea subsirurilor in BASIC se face cu:

s(n1 TO n2)

unde

1. s este un sir de caractere sau o variabila sir
2. n1,n2 sint numere intregi nenegative ce reprezinta ordinul caracterului de inceput, respectiv de sfirsit, din subsir. Daca n1>n2, rezultatul este sirul vid ("").

Daca nu se precizeaza inceputul si/sau sfirsitul subsirului se iau implicit 1, respectiv lungimea sirului.

Exemplul 1.

```
"abcde"(2 TO 5)="bcde"
"abcdef"( TO 5)="abcdef"(1 TO 5)="abcde"
"abcdef"(2 TO )="abcdef"(2 TO 6)="bcdef"
"abcdef"( TO )="abcdef"(1 TO 6)="abcdef"
"abcdef"(3)="abcdef"(3 TO 3)="c"
```

"abcdef"(5 TO 7) da mesaj de eroare deoarece sirul are numai sase caractere

```
"abcdef"(8 TO 7)=""
"abcdef"(1 TO 0)=""
```

Exemplul 2.

```
10 LET a$="able was !"
20 FOR n=1 TO 10
30 PRINT a$(n TO 10), a$((11-n) TO 10)
40 NEXT n
50 STOP
```


Exemplu

```
10 LET c$="acesta este un calculator HC-90"
20 LET c$(13 TO 25)="hc-90"
30 PRINT c$
```

Dupa executia programului pe ecran va apare mesajul:

Acesta este hc-90 HC-90

Daca intr-o atribuire sirul din dreapta contine mai putine caractere decit sint specificate in subsirul din stinga, atunci diferenta de lungime va fi completata cu blancuri.

Tablouri

Cuprins: Tablouri de numere si siruri
DIM

In limbajul BASIC al calculatorului HC 90 se pot defini variabile de tip tablou cu oricite dimensiuni. Elementele tabloului pot fi numere reale, caz in care numele variabilei este reprezentat pintr-o singura litera, sau de tip sir de caractere, numele variabilei fiind format dintr-o litera urmata de $. Inainte de a utiliza un tablou, trebuie rezervat spatiu in calculator pentru el; aceasta se realizeaza utilzind instructiunea DIM, a carei forma este

DIM m(n1,n2,..,nk)

unde

1. m - este numele unei variabile de tip tablou
2. n1,n2,..,nk - sint numerele maxime de componente corespunzatoare fiecarei dimensiuni a tabloului.

Printr-o comanda DIM poate fi definita numai o singura variabila de tip tablou. Aceasta instructiune are urmatorul efect:

- rezerva spatiul necesar tabloului definit
- initializeaza elementele tabloului cu 0
- sterge orice tablou care are acelasi nume cu variabila definita prin instructiunea curenta.

Se mentioneaza ca pot coexista un tablou si o variabila simpla cu acelasi nume, fara sa apara confuzii.

Sirurile dintr-un tablou difera de sirurile simple prin aceea ca au lungime fixa si asignarea lor este procusteana. Un alt mod de interpretare al unui tablou de siruri de caractere este ca tablou de caractere simple cu numarul dimensiunilor majorat cu 1 fata de cazul precedent. Un tablou de siruri si o variabila sir simpla nu pot avea acelasi nume (spre deosebire de cazul variabilelor numerice).

Pentru a defini un tablou a$ de 5 siruri, trebuie stabilita mai intii lungimea sirului - spre exemplu 10 caractere.

Linia:

```
DIM a$(5,10)
```

defineste 5*10=50 caractere, dar fiecare rind poate fi inter-

pretat ca un sir.

De exemplu a$(1) este format din:

a$(1,1) a$(1,2) ... a$(1,10)

Daca sint utilizate doua dimensiuni, se obtine un singur caracter, dar daca este omisa a doua dimensiune, atunci se obtine un sir cu lungime fixa. Astfel a$(2,7) e al saptelea caracter in sirul a$(2); o alta notatie a aceluiasi element este a$(2)(7).

Ultimul indice poate avea si forma unui selector de subsir. De exemplu, daca a$(2)="12345667890", atunci

a$(2,4 TO 8)= a$(2)(4 TO 8)="45678"

Se pot defini variabile de tip tablou de siruri de caractere cu o singura dimensiune; in acest caz variabila se comporta ca o variabila simpla cu exceptia faptului ca are totdeauna lungime fixa iar asignarea ei este procusteana.

Exemplu

```
DIM a$(10)
```

Initializarea variabilelor

Cuprins: READ,DATA,RESTORE

Introducerea constantelor intr-un program se face prin grupul de instructiuni READ, DATA si RESTORE. Forma generala a unei instructiuni READ este:

READ n1,n2,...

unde n1,n2,... este lista variabilelor care trebuiesc initializate, variabilele fiind separate prin virgula. Instructiunea READ lucreaza la fel cu instructiunea INPUT, exceptind faptul ca valorile variabilelor sint luate dintr-o instructiune DATA, nu de la terminal.

Fiecare instructiune DATA este o lista de expresii numerice sau de tip sir de caractere, separate prin virgula. Instructiunile DATA pot fi puse oriunde in program, ele comportindu-se ca o lista unica realizata prin concatenarea tuturor instructiunilor DATA din program (lista DATA).

Cind calculatorul citeste prima variabila cu READ, ei ii este asociata prima valoare din lista DATA, si asa mai departe. Daca se incearca citirea mai multor variabile decit numarul valorilor din lista DATA, atunci apare eroare.

Este posibil sa se faca salturi in lista DATA, utilizind instructiunea RESTORE. Forma instructinii este:

RESTORE n

Ea face ca instructiunea READ urmatoare sa citeasca datele de la o instructiune DATA aflata la linia "n" sau dupa aceasta. Daca "n" lipseste, se ia valoare implicita 1.

Exemplul 1.

```
10 READ a,b,c
20 PRINT a,b,c
30 DATA 10,20,30
40 STOP
```


Rezultatele programului vor fi:

```
 10    20        (a=10, b=20)
 30              (c=30)
```

Exemplul 2.

```
10 READ d$
20 PRINT "Data este:  ",d$
30 DATA "21 aprilie 1985"
```

Rezultatul acestui program este:

Data este: 21 aprilie 1985

Exemplu:

```
10 READ a,b
20 PRINT a,b
30 RESTORE 10
40 READ x,y,z
50 PRINT x,y,z
60 DATA 1,2,3
70 STOP
```

Rezultatele furnizate de acest program sint:

```
1     2        (a=1, b=2)
1     2        (x=1, y=2)
3              (z=3)
```

Operatii logice

Cuprins: =,<,>,<=,>=,<>
AND,OR,NOT

Operatiile aritmetice executate de calculator sint inmultirea, impartirea, adunarea si scaderea. Operatiile de adunare si scadere au prioritate egala dar mai mica decit a inmultirii si a impartirii .

Pentru sirurile de caractere s-a definit operatia de concatenare, notata cu "+".

Exemplul 1.

```
10 LET n$="Ionescu "
20 LET p$="Ana"
30 LET s$=n$+p$
40 PRINT s$
50 STOP
```

Programul prezentat va determina tiparirea pe ecran a textului

Ionescu Ana

care reprezinta valoarea variabilei s$.

Relatiile de ordine in multimea numerelor sint relatiile de egalitate si de inegalitate apelabile folosind notatiile "=", "<", ">","<=", ">=", "<>".

In multimea sirurilor de caractere relatia de ordine

folosita este ordonarea alfabetica, relatiile folosite fiin aceleasi ca la numere.

Pentru realizarea unor expresii complexe se pot utiliza s operatiile logice "OR", "AND" si "NOT" care admit operanzi d tip boolean. De exemplu instructiunea

```
IF a$="DA" AND x>0 THEN PRINT x
```

tipareste valoarea numarului "x" daca sint indeplinite simulta cele 2 conditii.

Similar se pot realiza expresii cu "OR" daca se dorest identificarea situatiei in care cel putin una dintre conditi este indeplinita. Operatia "NOT" produce ca rezultat inversu valorii argumentului sau.

Operatiile "OR", "AND","NOT" pot fi aplicate si unor argu mente numerice. Functiile definite astfel sint:

x AND y ia valoarea
- x , daca y e nenul
- 0 , daca y=0

x OR y ia valoarea
- 1 , daca y e nenul
- x , daca y=0

NOT x ia valoarea
- 0 , daca x e nenul
- 1 , daca x=0

In continuare vor fi prezentate operatiile recunoscute d limbajul BASIC in ordinea crescatoare a prioritatilor:

"OR"
"AND"
"NOT"
relatiile conditionale
"+", "-"
"*", "/"

Functii

Cuprins: !,PI,EXP,LN,SIN,COS,TAN,ASN,ACS,ATN
DEF,LEN,STR$,VAL,SGN,ABS,INT,SqR,FN

Functiile definite de calculator au prioritate mai mar decit operatiile. Daca in evaluarea unei expresii este necesar o alta ordine de executie a operatiilor si functiilor decit ce determinata de prioritatile lor, atunci se folosesc paranteze.

Functiile matematice definite in BASIC sint ridicarea l putere, functia exponentiala, functia logaritmica si func tiile trigonometrice.

Functia ridicare la putere "^" are prioritate mai mar decit inmultirea si impartirea. Ea necesita 2 operanzi dintr care primul este obligatoriu pozitiv. Intr-o insiruire de ri dicari la putere, ordinea evaluarii este de la stinga la dreapta ceea ce inseamna ca :

```
2^3^2=8^2=64
```


Functia EXP defineste functia exponentiala

```
EXP x=e^x
```

unde e=2,71...

Functia LN calculeaza logaritmul natural al argumentului. Ea poate fi utilizata la calculul unui logaritm in orice baza folosind formula:

```
LOGa x=LN x/ LN a
```

SIN, COS, TAN, ASN, ACS, ATN sint mnemonicele functiilor sinus, cosinus, tangenta, arcsinus, arccosinus si respectiv arc- tangenta.

Sistemul pune la dispozitia utilizatorului numarul "pi", ce poate fi apelat apasind tasta PI. Comanda PRINT PI tipareste valoarea numarului "pi".

Functiile descrise in continuare sint disponibile in modul de lucru extins. Actionarea simultana a tastelor CAPS SHIFT si SYMBOL SHIFT determina trecerea din modul "L" in modul "E".

Functia LEN da lungimea unui sir.

Exemplu

```
PRINT LEN "majuscule"
```

va determina tiparirea numarului 9.

Functia STR$, converteste numere in siruri. Argumen- tul este un numar, iar rezultatul este sirul care ar apare pe ecran daca numarul ar fi afisat cu PRINT. Se observa ca numele functiei se sfirseste cu "$" pentru a arata ca rezultatul ei este un sir.

Exemplu

```
LET a$=STR$ 1e2
```

Instructiunea de mai sus are acelasi efect cu

```
LET a$="100"
```

Comanda

```
PRINT LEN STR$ 100.000
```

produce raspunsul 3, deoarece STR$ 100.000="100".

Functia VAL converteste siruri de caractere in numere.

```
VAL "3.5"=3.5
```

Daca se aplica functiile STR$ si VAL asupra unui numar, totdeauna se va obtine numarul initial, pe cind daca se aplica VAL urmat de STR$ asupra unui sir de caractere nu se obtine totdeauna sirul initial. Evaluarea functiei VAL se face in 2 pasi:

1. argumentul este evaluat ca sir
2. ghilimelele sint indepartate si caracterele ramase sint evaluate ca numere.

Exemplu:

```
VAL "2"=2
VAL ("2"+"*3")=6
```

Alta functie similara lui VAL dar mai putin utilizata este VAL$. Si aceasta functie se evalueaza tot in 2 pasi; primul pas este la fel cu al functiei VAL, dar dupa inlaturarea ghilimelelor caracterele sint evaluate ca alt sir.

```
VAL$ """fructe"""="fructe"
```

Functia SGN aplicata asupra variabilei x are urmatoarea definitie:

1. 1, daca x>0
2. 0, daca x=0
3. -1, daca x<0

Functia ABS produce valoarea absoluta a numarului pe care-l are ca argument.

ABS -3.2=ABS 3.2=3.2

Functia INT furnizeaza partea intreaga a argumentului sau.

INT 3.9=3
INT -3.9=-4

Functia SQR calculeaza radacina patrata a argumentului sau care este un numar pozitiv.

SQR 0.25=0.5
SQR -4 ==>genereaza mesaj de eroare

Sistemul permite definirea de functii utilizator. Numele posibile pentru acestea sint FN urmat de o litera (daca rezultatul e un numar), sau FN urmat de o litera si $ (daca rezultatul e un sir). Obligatoriu argumentul trebuie sa fie inclus in paranteze. Definirea functiilor utilizator se face cu functia predefinita DEF. Definirea functiei de ridicare la patrat se poate face astfel:

DEF FN s(x)=x*x

Rotunjirea unui numar real la cel mai apropiat intreg poate fi facuta prin aplicarea functiei INT asupra argumentului marit cu 0.5

20 DEF FN r(x)=INT(X+0.5)

Exemplu:

```
10 LET x=0:  LET y=0:  LET a=10
20 DEF FN p(x,y)=a+x*y
30 DEF FN q()=a+x*y
40 PRINT FN p(2,3), FN q()
```

Cind este evaluata FN p(2,3),"a" are valoarea 10, deoarece e variabila libera, x are valoarea 2 deoarece este primul argument si y are valoarea 3 deoarece este al doilea argument. Rezultatul este: 10+2*3=16.

Cind este evaluata functia fara argumente FNq, a,x si y sint variabile libere si au valorile: 10, 0 respectiv 0. Raspunsul in acest caz este : 10+0*0=10.

Schimbind linia 20 cu

20 DEF FN p(x,y)=FN q()

de aceasta data FN p(2,3) va avea valoarea 10.

O functie poate avea pina la 26 argumente numerice si in acelasi timp pina la 26 argumente de tip sir de caractere.

Decizii

Cuprins: IF,THEN,STOP

Instructiunea care realizeaza luarea deciziilor este de forma

n IF conditie THEN comenzi

unde

1. "n" este numarul liniei
2. "comenzi" este o secventa de instructiuni care trebuie sa fie executata in cazul in care "conditia" este adevarata.
3. "conditie" este o relatie operationala care in urma evaluarii poate fi adevarata sau falsa. Daca conditia este adevarata, atunci se executa secventa de instructiuni scrisa dupa THEN. Altfel, programul executa instructiunile de pe linia urmatoare.

Cele mai simple conditii compara doua numere sau doua siruri de caractere. Ele pot testa daca doua numere sint egale sau daca unul este mai mare decit celalalt. Se poate testa si egalitatea a doua siruri de caractere, sau daca in ordinea alfabetica unul apare inaintea celuilalt.

Exemplu

```
10 REM Ghiciti numarul
20 INPUT a :  CLS
30 INPUT "Ghiciti numarul" , b
40 IF b=a THEN PRINT "Rezultat corect" :  STOP
50 IF b<A THEN PRINT "Prea mic!  Mai incearca  o
   data!"
60 IF b>a THEN PRINT "Prea mare!  Mai  incearca
   o data!"
70 GO TO 30
```

In acest program linia 40 compara variabilele a si b. Daca sint egale, programul este oprit cu comanda STOP. In partea de jos a ecranului apare mesajul

9 STOP statement, 40:3

care arata ca oprirea programului este cauzata de a treia instructiune din linia 40.

Linia 50 determina daca b este mai mic decit a, iar linia 60 opusul, adica daca b este mai mare decit a. Instructiunea CLS din linia 20 sterge ecranul si impiedica adversarul de joc sa vada ce numar s-a introdus.

Iteratii

Cuprins: FOR,NEXT,TO,STEP

In BASIC instructiunea de ciclare este FOR - NEXT. Forma generala a instructiunii FOR este:

FOR v=vi TO vf STEP p
corp ciclu
NEXT v

unde

1. "v" este o variabila contor specifica ciclului FOR - NEXT ; ea trebuie sa aiba numele format dintr-o singura litera.
2. "vi" este valoarea cu care'este initializat contorul ciclului
3. "vf" este valoarea maxima la care poate ajunge "v", deci "v" <= "vf" (s-a presupus ca "p" > 0).
4. "p" este marimea pasului; el reprezinta diferenta intre doua valori succesive ale contorului.
5. "corp ciclu" este secventa de instructiuni ce se repeta. "vi","vf" si "p" pot fi exprimate prin constante, variabile sau expresii de tip real.

In cazul in care "p" este negativ, regula de raminere in ciclu este "v" >= "vf".

Doua cicluri FOR - NEXT pot fi imbricate sau complet separate. Este gresita suprapunerea partiala a doua cicluri. De asemenea trebuie evitat saltul din exterior in interiorul unei bucle FOR - NEXT deoarece contorul nu poate fi initializat decit printr-o instructiune FOR. Pentru a fi siguri ca nu se fac salturi in interiorul unui ciclu se pot scrie toate instructiunile ciclului pe o singura linie (daca spatiul permite).

Exemple

```
10 FOR n=10 TO 1 STEP -1
20 PRINT n
30 NEXT n

50 FOR m=0 TO 6
60 FOR n=0 TO m STEP 1/2
70 PRINT m;":";n;" ";
80 NEXT n
90 PRINT
100 NEXT m

100 FOR m=0 TO 10:PRINT m:NEXT m

FOR n=0 TO 1 STEP 0:INPUT a:PRINT a:NEXT n
```

Aceasta comanda determina repetarea la infinit a instructiunii INPUT in modul de lucru imediat (deci nu prin program). Daca apare o eroare, comanda INPUT se pierde si deci pentru continuarea citirii trebuie rescrisa intreaga linie.

Subrutine

Cuprins: GOSUB,RETURN

Utilizarea subrutinelor este posibila prin utilizarea instructiunii GO SUB (go to subroutine-apel de subrutina) si RETURN (revenire din subrutina). Aceste instructiuni au forma:

GO SUB n

unde "n" este numarul primei linii din subrutina. Ea este asemanatoare instructiunii GO TO n , cu exceptia faptului ca in cazul instructiunii GO SUB este memorata adresa instructiunii, astfel incit dupa executarea subrutinei programul continua cu instructiunea urmatoare saltului la subrutina. Aceasta se realizeaza memorind numarul liniei si numarul instrutiunii din linie (care impreuna formeaza adresa de revenire) intr-o stiva.

Instructiunea RETURN ia adresa din virful stivei GO SUB si merge la instructiunea care ii urmeaza.

In BASIC subrutinele sint recursive.

Exemplu

```
10 INPUT a: CLS
20 INPUT "ghiciti numarul !",b
30 IF a=b THEN PRINT "corect !!!": STOP
40 IF a<b THEN GO SUB 90
50 IF a>b THEN GO SUB 90
60 GO TO 20
90 PRINT "Mai incearca o data !"
100 RETURN
```

Instructiunea GO TO este foarte importanta deoarece sistemul semnaleaza eroarea daca, in executie, intilneste un RETURN care nu a fost precedat de un GO SUB.

Generarea numerelor aleatoare

Cuprins: RND,RANDOMIZE

Generarea numerelor aleatoare se face cu functia predefinita RND. Ea nu este o functie complet aleatoare ci o functie periodica cu perioada suficient de mare (65535), astfel incit efectul de periodicitate poate fi neglijat. In cadrul unei perioade, numerele generate sint complet aleatoare. In anumite privinte, RND se comporta ca o functie fara argumente: efectueaza calcule si produce un rezultat. De fiecare data cind e utilizata, rezultatul sau este un numar aleator nou, cuprins intre 0 si 1 (uneori poate lua valoarea 0, dar niciodata 1). Daca se doreste ca numerele aleatoare sa fie intr-un anumit domeniu de valori se poate proceda ca in exemplele urmatoare:

5*RND	genereaza numere intre 0 si 5;
1.3+0.7*RND	produce numere intre 1.3 si 2;
1+INT(RND*6)	furnizeaza numere aleatoare intregi intre 1 si 6.

Exemplu

```
10 REM  Program de simulare a aruncarii zarurilor
20 CLS
30 FOR n=1 TO 2
40 PRINT 1+INT(RND*6);" ";
50 NEXT n
60 INPUT a$:GO TO 20
```

Linia 60 face sa fie generata o pereche de numere aleatoare după fiecare apasare a tastei CR.

Functia RANDOMIZE e utilizata pentru a face ca RND sa porneasca dintr-un punct definit al secventei de numere; argumentul sau este un numar intre 1 si 65535 care reprezinta numarul de ordine al viitorului apel al functiei RND. Efectul instructiunii RANDOMIZE se poate vedea in programul urmator.

```
10 RANDOMIZE 1
20 FOR n=1 to 5 :PRINT RND :NEXT n
30 PRINT:GO TO 10
```

Dupa fiecare executie a instructiunii RANDOMIZE 1 , RND va furniza o secventa de 5 numere ce incepe cu 0.0022735596, care este primul numar generat de functia RND (are numarul de ordine 1). RANDOMIZE poate fi folosit la testarea programelor ce contin functia RND, deoarece secventa numerelor aleatoare generate este mereu aceeasi.

RANDOMIZE, ca si RANDOMIZE 0 , are efect diferit de RANDOMIZE urmat de un numar. Aceasta instructiune utilizeaza timpul trecut de la punerea in functiune a calculatorului.
Programul

```
10 RANDOMIZE
20 PRINT RND:GO TO 10
```

determina tiparirea aceluiasi numar. Deoarece timpul de lucru al calculatorului a crescut cu aceeasi cantitate la fiecare executie a lui RANDOMIZE, urmatorul RND furnizeaza aproximativ acelasi rezultat.

Pentru a se obtine o secventa aleatoare se înlocuieste GO TO 10 cu GO TO 20.

Exemplu

Programul determina frecventa de aparitie a "capului" si a "pajurei" la aruncarea unei monezi.

```
10 LET cap=0:LET pajura=0
20 LET moneda=INT(RND*2)
30 IF moneda=0 THEN LET cap=cap+1
40 IF moneda=1 THEN LET pajura=pajura+1
50 PRINT cap;  ",";pajura
60 IF pajura>0 THEN PRINT cap/pajura;
70 PRINT:GO TO 20
```

Daca timpul de rulare este suficient de mare, raportul cap/pajura devine aproximativ 1, deoarece numerele aleatoare generate sint uniform repartizate in intervalul 0,1.

Setul de caractere

Cuprins: CODE,CHR$,POKE,PEEK,USR,BIN

Alfabetul utilizat de HC 90 cuprinde 256 caractere si fiecare are un cod intre 0 si 255. Caracterele pot fi simboluri simple sau cuvinte cheie ca PRINT, STOP, >, etc.

Pentru conversia intre coduri si caractere, limbajul poseda doua functii: CODE si CHR$. CODE se aplica unui sir si intoarce codul primului caracter al sirului (sau 0 daca sirul e vid).

CHR$ se aplica unui numar si produce caracterul ce are acel cod.

Setul de caractere este format din: caracterele ASCII, cuvinte cheie, caractere grafice definite de utilizator.

Un caracter se deseneaza pe o retea de 8*8 puncte, fiecarui punct corespunzindu-i un bit in memorie. Pentru programarea unui caracter definit de utilizator este necesara descrierea starii fiecarui punct al matricii prin care se reprezinta caracterul respectiv:

1. 0 corespunde unui punct alb
2. 1 corespunde unui punct negru

Pentru definirea caracterului se folosesc 8 instructiuni BIN. O instructiune BIN descrie o linie a caracterului, argumentul sau fiind format din 8 cifre binare.

Cele 8 numere sint memorate in 8 octeti care corespund aceluiasi caracter.

Instructiunea USR converteste un argument de tip sir in adresa din memorie a primului octet al caracterului definit de utilizator corespunzator argumentului. Argumentul trebuie sa fie un singur caracter; el poate fi graficul definit de utilizator sau litera corespunzatoare (majuscula sau minuscula).

POKE memoreaza un numar direct intr-o locatie de memorie, fara sa faca apel la mecanismele utilizate in mod obisnuit in BASIC. Opusul lui POKE este PEEK, care ne permite sa vizualizam continutul unei locatii de memorie, fara a-l modifica.

Pentru a defini caracterul grafic pi (care sa apara pe ecran la apasarea tastei P in mod grafic) se utilizeaza urmatoarea secventa de program:

```
10 FOR n=0 TO 7
20 INPUT  r:POKE USR "p"+n, r
30 NEXT n
```

Datele introduse vor fi (in ordinea prezentata):

```
BIN 00000000
BIN 00000000
BIN 00000010
BIN 00111100
BIN 01010100
BIN 00010100
BIN 00010100
BIN 00000000
```

Dupa introducerea datelor daca trecem in modul grafic si apasam tasta P vom obtine in loc de litera P simbolul grafic al numarului PI.

In cele ce urmeaza se prezinta modul de obtinere a cuvintelor cheie. Caracterele 0,..,31 sint caractere de control al modului de lucru. De exemplu CHR$6 realizeaza tabularea pe orizontala (efect similar unei virgule intr-o instructiune PRINT).

```
PRINT 1;  CHR$ 6;  2
```

are acelasi efect cu:

```
PRINT 1,2
```

si cu:

```
LET a$="1"+CHR$6+"2"
PRINT a$
```

CHR$8 determina mutarea cursorului inapoi cu o pozitie.

Exemplu:

```
PRINT "1234";  CHR$8;  "5"
```

tipareste:

```
1235
```

Initial a fost tparit 1234 apoi CHR$8 a intors cursorul inapoi cu o pozitie si a tiparit peste 4 urmatorul caracter in cazul nostru 5. Totul s-a petrecut atit de repede incit pe ecran am vazut rezultatul final in forma 1235.

CHR$13 muta cursorul la inceputul liniei urmatoare.

Utilizind codurile pentru caractere putem extinde conceptul de ordine alfabetica pentru a acoperi siruri ce contin orice caractere, nu numai litere, folosind in locul alfabetului uzual de 26 litere, alfabetul extins de 256 caractere (la codificarea caracterelor s-a avut in vedere ca ordinea crescatoare a codurilor atasate literelor sa coincida cu ordinea alfabetica).

Este prezentata mai departe o regula de gasire a ordinii in care se afla doua siruri. Mai intii se compara primele caractere. Daca sint diferite, unul dintre ele are codul mai mic decit celalalt si, deci, se poate decide care este ordinea alfabetica a sirurilor. Daca aceste coduri sint egale, se compara urmatoarele caractere.

Exemplu

```
5 LET b=BIN 01111100:LET c=BIN 00111000:LET d=BIN
  00010000
10 FOR n=1 TO 6:  READ p$:  REM 6 piese
20 FOR f=0 TO 7:  REM citeste piesele in octeti
30 READ a:  POKE USR p$+f,a
40 NEXT f
50 NEXT n
100 REM bishop
110 DATA "b", 0, 0, BIN 001001000, BIN 01000100
120 DATA BIN 01101100, c, b, 0
130 REM king
140 DATA "k", 0, d, c, d
150 DATA c, BIN 010001000, c, 0
160 REM rook
170 DATA "r", 0, BIN 01010100, b, c
180 DATA c,b,b,0
190 REM queen
200 DATA "q",0,BIN 01010100, BIN 00101000, d
210 DATA BIN 01101100, b, b, 0
220 REM pawn
230 DATA "p", b, 0, d, c
240 DATA c, d, b, 0
250 REM knight
260 DATA "n", 0, d, c, BIN 01111000
270 DATA BIN 00011000, c, b, 0
```

3.12 Grafice

Cuprins: PLOT,DRAW,CIRCLE,POINT

In acest capitol se prezinta trasarea desenelor. Partea utilizabila a ecranului are 22 de linii si 32 de coloane (22*32=704 pozitii de caractere). Fiecare pozitie de caracter e un patrat format din 8*8 puncte. Punctele se numesc pixeli (picture elements). Un pixel se specifica prin coordonatele sale. Coordonata "x" arata distanta fata de extrema stinga, iar coordonata "y" reprezinta distanta fata de baza ecranului. Coordonatele se scriu de obicei ca o pereche de numere, in paranteze. Astfel (0,0), (255,0), (0,175), (255,175) sint extremele stinga jos, dreapta jos, stinga sus, dreapta sus.

Instructiunea PLOT lucreaza cu coordonate absolute, x si y specifica un punct din spatiul de 256 X 176 pixeli ecran lasind libere ultimile doua rinduri ecran.

Instructiunea

```
PLOT      x,y
```

deseneaza punctul de coordonate x,y.

```
10 PLOT INT (RND *256), INT(RND *175):INPUTa$:GO TO 10
```

scrie aleator un punct pe ecran de fiecare data cind se actionea-

za CR. Programul urmator traseaza graficul functiei SIN pentru valori .ntre 0 si 2*pi.

```
10 FOR n=0 TO 255
20 PLOT n,88+80*SIN(n/128*pi)
30 NEXT n
```

Calculatorul deseneaza linii drepte, cercuri si portiuni de cerc utilizind instructiunile DRAW si CIRCLE. Cu

DRAW x,y

Spre deosebire de PLOT, x si y din instructiunea DRAW sint coordonate relative. Ele indica deplasarile pe x respectiv y care ne aduc din punctul initial corespunzator cursorului in punctul final dupa executia instructiunii DR..

Prin cursor grafic intelegem ultima pozitie atisa dupa executarea unei instructiuni grafice de tip PLOT, DRAW, CIRCLE.

Instructiunea DRAW cu doua argumente traseaza o linie dreapta.

Comenzile RUN, CLEAR, CLS si NEW il reseteaza cursorul grafic, aducindu-l pe pozitia (0,0).

DRAW determina lungimea si directia liniei. De remarcat ca argumentele unei instructiuni DRAW pot fi si negative.

```
PLOT 0,100:  DRAW 80,-35
PLOT 90,150:  DRAW 80,-35
```

Calculatorul HC 90 are facilitati pentru a desena in culori. Urmatorul program demonstreaza acest lucru:

```
10 BORDER 0:  PAPER 0:  INC 7:  CLS:  REM tot
   ecranul este negru
20 LET x1=0:  LET y1=0:  REM inceputul liniei
30 LET c=1:  REM prima culoare cu care se de-
   seneaza este albastru
40 LET x2=INT(RND*255):  LET y2=INT(RND*176):
   REM capatul liniei este aleator
50 DRAW INK c;  x2-x1,y2-y1
60 LET x1=x2:  let y1=y2:  REM urmatoarea linie
   incepe de unde s-a terminat precedenta
70 LET c=c+1:  IF c=8 THEN LET c=1:  REM  alta
   culoare
80 GO TO 40
```

Comenzile PAPER, INK, FLASH, BRIGHT, INVERSE, OVER pot apare in instructini PLOT sau DRAW in acelasi fel in care apar in PRINT si INPUT.

DRAW cu trei argumente permite si trasarea de portiuni de cercuri.Forma generala este

DRAW x,y,a

unde x,y sint coordonatele relative ale punctului final al liniei iar a este numarul de radiani corespunzator circumferintei. Cind a este pozitiv portiunea de cerc se traseaza in sens antiorar in timp ce, pentru a negativ se deseneaza in sens orar. Pentru a=pi se traseaza un semicerc, indiferent de valorile luate de x si y (raza este functie de punctul initial si de cel final):

```
10 PLOT 100,100:DRAW 50,50,pi
```

Trasarea cercurilor se face cu o comanda CIRCLE a carei

forma este

CIRCLE x,y,r

unde r este raza cercului iar (x,y) sint coordonatele centrului cercului. Ca si instructiunea PLOT si DRAW, si CIRCLE admite comenzi de modificare a culorii.

Functia POINT arata daca un pixel are asociata culoarea INK sau culoarea PAPER. Ea are doua argumente numerice care reprezinta coordonatele pixel-ului care trebuie sa fie inchis intre paranteze. Rezultatul este

1. 0 -daca punctul are culoarea fundalului (paper)
2. 1 -daca are culoarea INK.

```
CLS:  PRINT POINT (0,0):  PLOT 0,0:  PRINT  POINT(0,0)
```

Se scrie

```
PAPER 7:INK 0
```

Intr-o instructiune PLOT x,y , REVERSE si OVER afecteaza doar pixel-ul desemnat, nu si restul pozitiilor din caracter. Deoarece aceste comenzi sint in mod normal dezactivate (0), pentru a le activa (1), trebuiesc incluse intr-o comanda PLOT.

Se poate face ca punctul (x,y) sa ia culoarea "ink" prin PLOT x,y;

```
PLOT INVERSE 1;
```

face ca pixel-ul (x,y) sa ia culoarea fundalului

```
PLOT OVER 1;  x,y
```

inverseaza culoarea pixel-ului specificat.

```
PLOT INVERSE 1;  OVER 1;  x,y
```

lasa pixel-ul nemodificat dar schimba pozitia de tiparire.

Alt exemplu de utilizare al instructiunii OVER este urmatorul:

-se umple ecranul scriind negru pe alb si apoi se tasteaza:

```
PLOT 0,0:  DRAW OVER 1,255,175
```

-se traseaza astfel o linie (cu intreruperi acolo unde traverseaza caracterele tiparite pe ecran).

-reexecutind comanda, linia trasata anterior o sa dispara.

Avantajul instructiunii OVER este ca permite sa se deseneze si apoi sa se stearga desenele fara a afecta ce se afla anterior pe ecran.

Utilizind programul

```
PLOT  0,0:  DRAW 255,175
PLOT 0,0:  DRAW INVERSE 1;  255,175
```

se constata ca aceasta comanda sterge si partile din caracterele tiparite anterior. Daca se scrie o linie cu:

```
PLOT 0,0:  DRAW OVER 1;  250,175
```

se constata ca ea nu va putea fi stearsa cu:

```
DRAW OVER 1;-250,-175
```

deoarece parcurgerea dreptei intr-un sens si in celalalt nu se face exact prin aceleasi puncte. O linie se sterge pe aceeasi directie si in acelasi sens in care a fost trasata.

Pentru a extinde gama de culori se amesteca doua culori de baza pe un singur patrat, folosind un caracter grafic definit de utilizator. Programul urmator defineste un caracter grafic echivalent unei table de sah.

```
1000 FOR n=0 TO 6 STEP 2
1010 POKE  USR  "a"+n,  BIN  01010101: POKE USR
     "a"+n+1, BIN 10101010
1020 NEXT n
```

Instructiuni de intrare-iesire

Cuprins: PRINT,INPUT
Utilizarea separatorilor :,;,TAB,AT,LINE,CLS

Expresiile folosite pentru a tipari valori cu instructiunea PRINT sint numite elementele instructiunii si sint separate intre ele cu virgula sau punct si virgula (separatori). Un element al instructiunii PRINT poate lipsi si in acest caz pot apare 2 virgule, una dupa alta.

Exista 2 elemente ale instructiunii PRINT care servesc la pozitionarea cursorului in vederea tiparirii. Acestea sint AT si TAB.

AT linie,coloana

deplaseaza cursorul (locul unde va fi tiparit urmatorul element) la linia si la coloana specificate. Liniile sint numerotate de la 0 la 21 (de sus in jos) si coloanele de la 0 la 31 (de la stinga la dreapta).

Exemplu

PRINT AT 11,16;"*"

imprima un asterisc in centrul ecranului. Instructiunea

TAB coloana

deplaseaza cursorul in coloana specificata. TAB determina deplasarea pe aceeasi linie pe care se gaseste cursorul, exceptind cazul cind pozitia de tiparire specificata se afla inaintea pozitiei de tiparire actuala; in aceasta situatie se face o deplasare la linia urmatoare.

Observatie: calculatorul considera coloanele din instructiunea TAB "modulo 32" (i.e. TAB 33 este echivalent cu TAB 1).

Exemplul de mai jos arata cum se poate tipari inceputul paginii 1 a unei carti:

```
PRINT  TAB  30;1;TAB 12;   "Index";    AT   3,1;
"Capitol";  TAB 24;  "Pagina"
```

Un exemplu din care rezulta reducerea modulo 32 a numarului din instructiunea TAB este urmatorul:

```
10 FOR n=0 TO 20
20 PRINT TAB 8*n;n;
30 NEXT n
```

De retinut urmatoarele observatii:

1.Elementele de tiparire care urmeaza instructiunilor TAB sau AT sint de obicei terminate cu ";". Daca s-ar folosi ";" sau nimic, cursorul, dupa ce este pozitionat, se deplaseaza.

2.Liniile 22 si 23 ale ecranului nu pot fi folosite pentru tiparire. Ele sint rezervate pentru comenzi, pentru citirea datelor, mesaje, etc.

3.Tiparind cu AT intr-o pozitie deja scrisa, ultima tiparire o anuleaza pe precedenta.

CLS sterge tot ecranul, functie care mai este realizata si de comenzile CLEAR si RUN (care mai executa si alte functii).Cind calculatorul,in timp ce tipareste,ajunge la ultima linie a ecranului,executa "scrolling" anulind prima linie.

Exemplu:

CLS: FOR n=1 TO 22: PRINT n: NEXT n

si apoi,

PRINT 99

de mai multe ori.

In timpul tiparirii, dupa ce calculatorul a umplut complet ecranul, se opreste scriind in partea de jos:

scroll ?

Se raspunde cu "y" sau "n".

Instructiunea INPUT

O linie de INPUT este compusa dintr-o serie de elemente si de separatori care au aceeasi functie ca intr-o linie de PRINT. INPUT considera orice element care incepe cu o litera ca pe o variabila asignabila (careia urmeaza sa i se introduca valoarea de la tastatura). Instructiunea INPUT poate tipari si mesaje; pentru a tipari un sir de caractere este suficienta introducerea acestuia intre ghilimele. Daca contine si valori de variabile, mesajul se inchide intre paranteze.

Daca se doreste citirea unei variabile de tip sir de caractere, a$, pe ecran apare caracterul ghilimele. Daca aceasta variabila trebuie sa ia valoarea unei alte variabile de tip sir definita in program, b$, aceasta se face prin stergerea ghilimelelor si introducerea numelui variabilei (b$).

Toate elementele instructiunii PRINT care nu sint supuse acestor reguli pot fi elemente ale instructiunii INPUT.

Exemple:

```
LET virsta mea=INT( RND*100):  INPUT  ("Eu  am";
virsta mea; "ani." ); "citi ani ai ?"; virsta ta
```

Variabila "virsta mea" este continuta intre paranteze, deci valoarea sa se tipareste, in timp ce variabila "virsta ta" nu este intre paranteze, si deci valoarea sa se citeste de la tastatura.

O alta modalitate de citire a variabilelor sir consta in scrierea cuvintului cheie LINE dupa INPUT si inaintea variabilei sir de citit:

INPUT LINE a$

In acest caz calculatorul nu va tipari ghilimelele, care in mod normal sint tiparite cind se asteapta introducerea unei variabile sir, chiar daca se comporta ca si cum ar fi fost. Astfel, scriind carte ca variabila de intrare, a$ va lua valoarea "carte". Deoarece ghilimelele nu sint tiparite, nu este posibila introducerea altui sir. De notat ca LINE nu poate fi folosit pentru variabile numerice.

Caracterele de control CHR$22 si CHR$23 functioneaza aproape similar lui AT si TAB. Caracterul de control pentru AT este CHR$22. Primul caracter care il urmeaza specifica numarul de linie, iar al doilea numarul coloanei, astfel ca

```
        PRINT CHR$22 +CHR$1 +CHR$ c;
```
este analog lui
```
        PRINT AT 1,c;
```
CHR$1 si CHR$ c(c=13) in mod normal au alta semnificatie, pe care insa si-o pierd cind urmeaza dupa CHR$22.

Caracterul de control echivalent lui TAB este CHR$23 si cele doua caractere care-l urmeaza sint folosite pentru a indica un numar cuprins intre 0 si 65535 care specifica numarul de TAB ca si argumentul unei instructiuni TAB.
```
        PRINT CHR$ 23+CHR$ a+CHR$ b
```
este echivalent lui
```
        PRINT TAB a+256*b
```
Daca nu se doreste afisarea mesajului "scroll ?" la sfirsitul fiecarui ecran, se poate folosi:
```
        POKE 23692,255
```
din cind in cind. Dupa aceasta linie calculatorul inhiba mesajul "scroll ?" pentru urmatoarele 255 linii.

Culori

Cuprins: PAPER,INK,FLASH,INVERSE,OVER,BORDER,ATTR

Calculatorul HC 90 are facilitati color. El foloseste 8 culori (numerotate de la 0 la 7). Lista culorilor in ordinea in care sint pe tastele numerice este urmatoarea:

0 - negru
1 - albastru
2 - rosu
3 - purpura (magenta)
4 - verde
5 - albastru deschis
6 - galben
7 - alb

Intr-un televizor alb-negru aceste numere corespund unor tonuri de gri ordonate de la inchis spre deschis.

Orice caracter are asociate 2 culori: culoarea caracterului propriu-zis si culoarea fondului (vezi subcapitolul Setul de caractere). La pornirea calculatorului, sistemul lucreaza in alb - negru, cu caractere negre pe fond alb. Tiparirea poate fi facuta normal, dar exista si posibilitatea sa apara pe ecran pilpiind (flash). Pilpiirea se obtine inversind continuu culoarea caracterului cu culoarea fondului. Deoarece atributele de culoare si pilpiire sint asociate caracterelor (deci matricilor de 64 puncte), nu este posibil ca intr-un caracter sa fie mai mult de doua culori. Valorile acestor atribute pot fi modificate cu instructiunile INK, PAPER si FLASH. Forma acestor instructiuni este:

```
        PAPER n
        INK n
        FLASH m
```
unde

1. n este un numar cuprins intre 0 si 7
2. m este un numar binar (0 pentru inactiv si 1 pentru activ).

Pentru ilustrarea modului de folosire al instructiunilor prezentate se propune programul:

```
20 FOR n=1 TO 10
30 FOR c=0 TO 7
40 PAPER c:  PRINT " ";:REM spatii colorate
50 NEXT c:  NEXT n
60 PAPER 7
70 FOR c=0 TO 3
80 INK c:  PRINT c;" ";
90 NEXT c:  PAPER 0
100 FOR c=4 TO 7
110 INK c:  PRINT c;" ";
120 NEXT c
130 PAPER 7:  INK 0
```

In afara de aceste valori de argumente a caror semnificatie a fost deja prezentata, mai pot fi folosite valorile 8 si 9. 8 poate fi folosit ca argument pentru toate cele 4 comenzi si semnifica transparenta, fapt ce nu altereaza atributele pozitiei la tiparirea unui caracter.De exemplu
```
        PAPER 8
```
face ca la tiparirea unui caracter, culoarea fondului sa fie aceeasi cu a caracterului tiparit anterior. 9 poate fi folosit numai cu comenzile PAPER si INK si indica contrastul. Culoarea "cernelii" sau a "hirtiei" (fundalului), in functie de comanda utilizata, este facuta sa contrasteze cu cealalta, punind alb pe o culoare inchisa (negru , albastru, rosu, magenta) si negru pe o culoare deschisa (verde, bleu, galben, alb).
```
        INK 9:  FOR c=0 TO 7:  PAPER c:  PRINT c:  NEXT c
```
Rulind programul
```
        INK 9:  PAPER 8:  PRINT AT 0,0;  FOR n=1 TO  1000:
        PRINT n:  NEXT n
```
dupa primul program din acest paragraf, culoarea cernelii este facuta mereu sa contrasteze cu vechea culoare pe care o avea fundalul in fiecare pozitie. Comanda
```
        INVERSE 1
```
inverseaza fundalul cu cerneala pentru caracterul specificat.

Comanda
```
        OVER 1
```
realizeaza supratiparirea. In mod obisnuit, cind ceva este scris intr-o pozitie de caracter sterge complet ce era scris inainte; de data aceasta noul caracter va fi doar adaugat. Acest lucru este util in scrierea caracterelor compuse, cum ar fi literele cu accente. Trebuie utilizat in acest scop caracterul de control CHR$8 pentru intoarcerea cu un spatiu.

Exista o alta posibilitate de a utiliza INK, PAPER, FLASH. Pot apare in PRINT urmate de ";" si fac exact acelasi lucru pe care l-ar face cind sint utilizate independent, exceptind faptul ca efectul lor este numai temporar.

Astfel daca se ruleaza:

```
PRINT PAPER 6;  "x";:  PRINT "y"
```

numai x va fi pe fond galben.

INK si celelalte comenzi nu afecteaza culorile partii din jos a ecranului. Aceasta foloseste culoarea marginii drept culoare a fundalului si codul 9 pentru a contrasta culoarea cernelii. Nu are posibilitatea de pilpiire si este cu luminozitate normala.

Marginea poate lua oricare din cele 8 culori (0-7) cu comanda

```
BORDER culoare
```

Se pot schimba culorile mesajului scris pe ecran cu comanda INPUT, inserind in aceasta comanda INK, PAPER, etc, ca si in cazul comenzii PRINT. Efectul lor este activ numai asupra comenzii urmatoare:

```
ink" prin
PLOT x,y;
PLOT INVERSE 1;
```

face ca pixel-ul (x,y) sa ia culoarea fundalului

```
PLOT OVER 1;  x,TAB (vezi capitolul Instructiuni de int
```

iesire).

```
CHR$16 --> INK
CHR$17 --> PAPER
CHR$18 --> FLASH
CHR$20 --> INVERSE
CHR$21 --> OVER
```

Aceste caractere de control sint urmate de un caracter care indica culoarea prin intermediul codului sau. De exemplu:

```
PRINT CHR$16 + CHR$9 ;  ...
```

are acelasi efect cu:

```
PRINT INK 9;  ...
```

Functia ATTR are forma:

```
ATTR (linie,coloana)
```

Rezultatul este un numar care arata atributele pentru caracterul aflat la linia si coloana precizata. Numarul trimis este suma a patru numere,conform schemei:

1. 128 - daca pozitia pilpiie , 0 daca este stabila
2. 64 - daca pozitia este stralucitoare , 0 daca este normala
3. 8*n - n=codul fundalului
4. m - m=codul cernelii

Exemplu: Pentru o pozitie pilpiitoare, normala, cu fundal galben si cerneala albastra se obtine:

```
128+0+8*6+1=177
```


Cuprins: PAUSE,INKEY$,PEEK

Pentru a realiza o pauza in program in timpul careia nu se desfasoara nici o operatie se foloseste comanda:

```
PAUSE n
```

care opreste executia programului mentinind activ display-ul pe durata a n perioade de baleiaj ale ecranului (20 ms pentru fiecare ecran); n poate lua valoarea maxima 65535, careia ii corespunde o pauza de aproximativ 22 minute. Daca n=0 se opreste definitiv.

O pauza obtinuta in acest mod poate fi scurtata apasind orice tasta (cu exceptia lui SPACE si CAPS SHIFT care produce BREAK).

Programul urmator deseneaza cadranul unui ceas pe care se misca secundarul:

```
10 REM Mai intii e desenat cadranul.
20 FOR n=1 TO 12
30 PRINT AT 10-10*COS( n/PI), 16+10*SIN( n/PI)
40 NEXT n
50 REM Se porneste ceasul.
60 FOR t=0 TO 200000; :REM t e timpul in secunde
70 LET a=t/30*PI:  REM a este unghiul secundaru-
   lui in radiani
80 LET sx=80*SIN( a):  LET sy=80*COS( a)
200 PLOT 128,88:  DRAW OVER 1;  sx, sy:  REM  Se
    deseneaza secundarul
210 PAUSE 42
220 PLOT 128,88:  DRAW OVER 1;  sx, sy:  REM  Se
    sterge secundarul
230 NEXT t
```

Cu linia 210 se marcheaza trecerea unei secunde; s-a folosit n=42 si nu n=50 deoarece calculatorul foloseste un timp pentru scrierea liniilor ciclului FOR - NEXT; linia 210 opreste calculatorul doar pentru timpul care mai ramine.

O temporizare mai precisa se poate realiza citind continutul anumitor locatii de memorie cu PEEK. Expresia urmatoare

```
(65536 *PEEK 23674+ 256*PEEK 23673+ PEEK 23672)/50
```

da numarul de secunde scurse de la aprinderea calculatorului pina la 3 zile si 21 ore, dupa care se reseteaza. Programul unui ceas mai precis este dat in continuare:

```
10 REM Se deseneaza cadranul
20 FOR n=1 TO 12
30 PRINT AT 10-10*cos(n/6*pi),16+10*SIN(n/6*PI);n
40 NEXT n
50 DEF FNt()= INT(65536* PEEK 23674+ 256*  PEEK
   23673+ PEEK 23672)/50:  REM Numarul de secunde de
   la inceput
100 REM se porneste ceasul
110 LET t1=FNt()
120 LET a=t1/30*PI:  REM a este unghiul in radi-
```

```
    ani
130 LET sx=72* SIN a:  LET sy=72*COS a
140 PLOT 131,91:  DRAW OVER 1;  sx;  sy:  REM
    Se deseneaza secundarul
200 LET t=FNt()
210 IF t=t1 THEN GO TO 200.
220 PLOT 131,91:  DRAW OVER 1;  sx;  sy:  REM Se
    sterge vechiul secundar
230 LET t1=t:  GO TO 120
```

Acest ceas se opreste temporar de cite ori se executa BEEP ori se utilizeaza imprimanta, casetofonul. Numerele PEEK 23674, PEEK 23673 si PEEK 23672 sint folosite pentru a numara in incremente de 20 ms. Fiecare variaza de la 0 la 255, dupa care se reincepe. Cel mai rapid se incrementeaza locatia 23672 (cu 1 la fiecare 20 ms); cind se trece de la 255 la 0, locatia 23673 se incrementeaza cu 1; analog pentru 23674. Presupunind ca cele 3 numere sint 0 (pentru PEEK 23674), 255 (pentru PEEK 23673) si 255 (pentru PEEK 23672), au trecut deci circa 21 minute de la pornirea calculatorului. Expresia devine

(65536*0+256*255+255)/50=1310.7

Pentru a pozitiona ceasul pe ora 10 se procedeaza astfel:

10*60 *60 *50=1800000= 65536*27 +256*119 +64

si se memoreaza numerele 27, 119 si 64 cu

POKE 23674,27: POKE 23673, 119: POKE 23672,64

Functia INKEY$, fara argument, da caracterul apasat pe tasta in momentul apelarii sale. Cu programul urmator calculatorul devine o masina de scris:

```
10 IF INKEY$>"" THEN GO TO 10
20 IF INKEY$="" THEN GO TO 20
30 PRINT INKEY$;
40 GO TO 10
```

Linia 10 asteapta sa se elibereze ultima tasta apasata; linia 20 asteapta apasarea uneia noi. Spre deosebire de INPUT, INKEY$ nu asteapta apasarea lui CR sau a unei taste.

3.16 Memoria

Cuprins: CLEAR

Fiecarui octet ii este asociata o adresa care este un numar intre 0 si FFFFH.

Memoria este impartita in trei zone distincte:

1. 0 - 4000H zona ROM
   in aceasta zona se gaseste memoria ROM in care este inregistrat interpretorul BASIC.
2. 4000H - 7FFFH zona RAM video
   in aceasta zona se gaseste memoria video cit si o parte din memoria RAM de program

3. 8000H - FFFFH zona RAM suplimentar
   aceasta zona nu este neaparat necesara. Ea este folosita pentru marirea capacitatii de memorie. Ea difera de zona video printr-un timp de acces mai mic



Continutul memoriei poate fi vizualizat cu functia PEEK care are ca argument o adresa. Exemplul urmator vizualizeaza primii 21 octeti din memoria ROM si adresele lor:

```
10 PRINT "Adresa";  TAB 10;  "Octet"
20 FOR a=0 TO 20
30 PRINT a TAB 10;  PEEK a
40 NEXT a
```

Schimbarea continutului memoriei RAM se poate face cu instructiunea POKE, care are forma:

POKE adresa, continut nou

unde "adresa" si "continut nou" sint expresii numerice.

POKE 31000, 57

determina incarcarea valorii 57 la adresa 31000. Cu

PRINT PEEK 31000

se va tipari 57. "Continut nou" trebuie sa aiba valoarea intre -255 si 255. Daca e numar negativ, se aduna 256.

De importanta pentru utilizator este organizarea memoriei RAM. Memoria este impartita in zone specifice stocarii unui anumit gen de informatie. Zonele sint suficient de mari pentru ca informatia continuta actualmente sa poata fi reorganizata daca se insereaza ceva intr-un anumit punct (de exemplu prin adaugarea unei linii de program sau a unei variabile). La inserare, spatiul necesar este creat prin mutarea in sus a tot ce se afla deasupra. Daca se sterge informatie, atunci totul este mutat in jos.



```
16384      22528        23296        23552        23734     CHANS

 ---------------------------------------------------------------
 |  Informatii  | 80H |  Program  |  Variabile  | 80H |
 |   de canal   |     |   BASIC   |             |     |
 ---------------------------------------------------------------
 ^                    ^           ^             ^
 CHANS                PROG        VARS          E-LINE

 ---------------------------------------------------------------
 | Comanda sau linia program |NL| 80h| Date in |NL| Spatiu de    |
 |  in curs de introducere   |  |    |  INPUT  |  | lucru temporar|
 ---------------------------------------------------------------
 ^                            ^                  ^
 E-LINE                       WORKSP             STKBOT

 ---------------------------------------------------------------
 |    Stiva    |Nefolosit| Stiva| Stiva | ?|3EH| Caractere grafice |
 | calculator  |         | PROC.| GOSUB |  |   |definite de utiliz.|
 ---------------------------------------------------------------
 ^             ^                        ^    ^                 ^
 STKBOT        STKEND                   RAMTOP UDG             P-RAMT
```

Variabilele sistem (PROG, CHANS, VARS, ELINE, etc) contin diferite informatii necesare pentru gestiunea interna a memoriei. Ele indica limitele pentru diverse zone de memorie . Ele nu sint variabile BASIC si deci nu pot fi recunoscute de calculator.

Fisierul display stocheaza imaginea televizorului. In loc de PEEK si POKE, pentru imaginea display-ului se pot utiliza SCREEN$ si PRINT AT sau PLOT si POINT.

Atributele sint culorile, etc pentru fiecare pozitie de caracter (se afla cu instructiunea ATTR). Ele sint stocate linie cu linie in ordinea dorita.

Buffer-ul imprimantei stocheaza caracterele destinate imprimantei.

Informatiile de canal sint necesare cind se lucreaza cu dispozitive de intrare - iesire. Si lucrul cu tastatura necesita aceasta zona deoarece partea de jos a ecranului functioneaza ca un port de intrare, in timp ce restul ecranului se comporta ca un port de iesire.

Orice linie de comanda are forma:

```
 ---------------------------------------------------------------
 | 2 bytes | 2 bytes |                          > >  | 00001101 |
 ---------------------------------------------------------------
      n         m              t                          e
```


unde

1. n - este numarul liniei curente
2. m - este lungimea textului + CR
3. t - este textul liniei
4. e - este codul caracterului CR

Modul de memorare al variabilelor numerice este:

```
 ---------------------------------------------------------------
 |       Nume        |      Exp      |      Mantisa          |
 ---------------------------------------------------------------
```

unde

1. Nume - este un numar de octeti egal cu numarul de caractere ce formeaza identificatorul variabilei
2. Exp - este un octet ce contine exponetul numarului
3. Mantisa - este un grup de 4 octeti ce contine mantisa numarului. Bitul cel mai semnificativ al primului octet este bitul de semn.

Producerea sunetelor

Cuprins: BEEP

Pentru producerea sunetelor, se foloseste instructiunea:

BEEP d,i

unde:

1. d - este o expresie numerica ce indica durata in secunde a sunetului respectiv
2. i - este o expresie numerica ce reprezinta inaltimea sunetului, masurat in semitonuri relativ la DO central.

Pentru a transcrie muzica este indicat sa se scrie pe marginea fiecarui spatiu si linii a portativului inaltimea corespunzatoare, tinind cont de armura cheii.

Exemplu:

```
10 PRINT "Frere Gustav"
20 BEEP 1,0:BEEP 1,2:BEEP .5,3:BEEP .5,2:BEEP 1,0
30 BEEP 1,0:BEEP 1,2:BEEP .5,3:BEEP .5,2:BEEP 1,0
40 BEEP 1,3:BEEP 1,5:BEEP 2,7
50 BEEP 1,3:BEEP 1,5:BEEP 2,7
60 BEEP .75,7:BEEP .25,8:BEEP .5,7:BEEP .5,5:BEEP .5,3:
   BEEP .5,2:BEEP1,0
70 BEEP .75,7:BEEP .25,8:BEEP .5,7:BEEP .5,5:BEEP .5,3:
   BEEP .5,2:BEEP1,0
80 BEEP 1,0:BEEP 1,-5:BEEP 2,0
90 BEEP 1,0:BEEP 1,-5:BEEP 2,0
```

Pentru alcatuirea programului s-a procedat dupa cum urmeaza:

1. s-au adaugat mai intii deasupra si dedesubt cite o linie de referinta
2. se numeroteaza liniile si spatiile, observind ca mi bemol din armura cheii afecteaza nu numai mi de sus (coborindu-l de la 16 la 15) cit si mi de jos (coborindu-l de la 4 la 3)

Pentru a schmba cheia partiturii, trebuie sa se adune la inaltimea fiecarei note o variabila (de exemplu "Cheie") careia trebuie sa i se atribuie valoarea adecvata inaintea executiei piesei.

Linia 20 a programului devine

```
20 BEEP 1, C + 0:BEEP1
```

In acest exemplu variabila cheie C trebuie sa aiba valoarea 0 pentru DO minor, 2 pentru RE minor, 12 pentru DO minor in octava superioara, etc.

Cu acest sistem este posibila acordarea calculatorului cu un alt instrument, folosind valori zecimale pentru variabile "Cheie". De asemenea, este posibil sa se execute piese cu viteze diferite. In exemplul dat "o patrime" a fost programata sa dureze o secunda. Daca se introduce o variabila "PATRIME" p analog cu "Cheie" c, linia 20 devine:

```
20   BEEP  p,  c + 0: BEEP   p, c + 2:
     BEEP  p/2, c +3: BEEP p/2, c + 2:
     BEEP p, c + 0
```

In acest fel este posibila executia aceluiasi program in orice cheie, cu orice acordare.

Programul de mai jos:

```
FOR n=0 TO 1000:  BEEP 0.5 , n:  NEXT n
```

va produce note din ce in ce mai acute, pina la limita posibilitatilor calculatorului, cind acesta va tipari mesajul:

```
B integer out of range
```

Tiparind n se obtine inaltimea notei celei mai acute care poate fi produsa. Procedeul poate fi repetat pentru notele joase.

Sunetele din gama medie sint cele mai potrivite pentru a fi redate.

Utilizarea codului masina

Cuprins: USR

Calculatorul HC 90 poate fi dotat cu un asamblor inregistrat pe caseta. Introducerea programului scris in limbaj masina (functie executata in general de asamblor) se face in general cu specificarea adresei de inceput (cel mai bine este ca aceasta adresa sa se afle intre zona BASIC si zona caracterelor grafice

definite de utilizator).

La pornirea unui calculator HC 90 inceputul memoriei RAM, RAMTOP, se afla la adresa 65367



Se poate deplasa RAMTOP cu comanda CLEAR 65267 obtinindu-se neutilizarea de catre BASIC a 100 octeti incepind cu adresa65267.



Pentru a insera codurile obiect in memorie, se poate utiliza un program de genul:

```
10 LET a=65267
20 READ n:  POKE a,n
30 LET a=a+1:  GO TO 20
40 DATA 1,99,0,201
```

care introduce programul

```
LD BC,99
RET
```

ranspus in cod masina ca:
1, 99, 0 (pentru LD bc,99) si 201 (pentru RET).
Cind se termina cei 4 octeti specificati, apare mesajul:
E Out of DATA
Rularea programului introdus in cod masina se face cu instructiunea:
USR adresa de inceput
In exemplul de mai sus, cu:
PRINT USR 65267
se tipareste valoarea 99 din perechea de registre BC.
Adresa de revenire in BASIC este culeasa din stiva de instructiunea Z80 RET.
Calculatorul HC 90 are scoase in exterior magisralele de date, de adrese si de control prin intermediul unui conector de extensie.
Un program in limbaj masina poate fi memorat ca o informatie de tip byte; deci cu:
SAVE "nume" CODE adresa, lungime
unde adresa este adresa in zecimal de la care este memorat programul in cod masina iar lungimea este numarul de octeti pe care il ocupa acest program. In exemplul de mai sus :
SAVE "test" CODE 65267,4

Un program in limbaj de asamblare nu se poate lansa automat, odata incarcat; el poate fi insa lansat de un program in BASIC ca in exemplul:

```
10 LOAD "" CODE 65267,4
20 PRINT USR 65267
```

Dupa aceasta se executa:
SAVE "nume" LINE 10
si apoi
SAVE "test" CODE 65267,4
Rebobinind caseta si scriind:
LOAD "nume"
se incarca si se executa programul BASIC care, la rindul sau, va apela programul in limbaj masina.

Utilizarea porturilor de intrare/iesire

Cuprins: IN,OUT

Calculatorul HC 90 dispune de 65536 adrese de memorie de tip RAM si ROM organizate pe opt biti. El poate sa scrie cuvinte in memoria de tip RAM si poate sa citeasca cuvinte din memoriile de tip RAM si ROM. Analog sint 65536 porturi de intrare iesire. Aceste porturi sint folosite de procesor pentru a comunica cu exteriorul. Instructiunile sint:
IN adresa port
care preia octetul citit de la acel port;
OUT adresa port, valoare
inscrie valoarea octetului in portul de adresa specificat. Exista un ansamblu de adrese de intrare care citeste tastatura si conectorul de casetofon. Tastatura este impartita in 8 grupe de 5 taste fiecare. Lista porturilor utilizate este:
IN 65278 citeste grupul CAPS SHIFT - v
Aceste adrese sint 254+256*(255-2^n) cu n=0,...,7
Bitii d0,..,d4 sint asociati celor 5 taste din grupul specificata. D6 este asociat conectorului de casetofon.
Portul de iesire cu adresa 254 controleaza difuzorul (D4), conectorul de casetofon(D3) si determina culoarea chenarului (D2, D1, D0). Portul de adresa 251 controleaza imprimanta inscriere si citire; la citire verifica daca imprimanta este gata sa imprime o noua linie si la scriere trimite linia care trebuie sa fie tiparita. Porturile de adrese 254, 247 si 239 sint folosite pentru echipamentele suplimentare (capitolul Alte periferice).

Inregistrarea pe caseta

Cuprins: SAVE,VERIFY,LOAD,MERGE

Calculatorul HC 90 are posibilitatea sa inregistreze programe de pe banda magnetica cu orice cesetofon.

Conectarea calculatorului la casetofon se face cu ajutorul unui cablu special.

Pentru a memora un program pe banda, acesta trebuie sa primeasca un nume compus din maximum 10 caractere, litere si/sau cifre. Comanda este:

Save "nume"

Calculatorul raspunde cu mesajul:

Start tape then press any key.

La terminarea inregistrarii apare mesajul:

0 OK.

Pentru verificare se regleaza volumul casetofonului la nivel mediu si se conecteaza cablul ; se pozitioneaza banda in punctul in care a inceput inregistrarea. Comanda este:

VERIFY "nume"

In acest fel se verifica daca programul si variabilele inregistrate pe caseta sint identice cu cele din memoria calculatorului. Daca programul a fost inregistrat si chemat corect, pe ecran apare

Program "nume"

(In timpul cautarii programului specificat, calculatorul tipareste numele tuturor programelor pe care le intilneste) si la sfirsit mesajul

0 OK

in cazul unei erori de inregistrare (eroare ce apare la VERIFY) se afiseaza mesajul:

R Tape loading error

se incearca o noua inregistrare. Incarcarea unui program memorat pe caseta se face cu comanda:

LOAD "nume"

Aceasta comanda sterge vechiul program (si variabilele sale) din calculator inainte de a incarca unul nou.

LOAD ""

fara a fi urmat de un nume de program incarca primul program gasit pe caseta.

Comanda MERGE incarca un program inregistrat pe caseta in memoria calculatorului, dar spre deosebire de comanda LOAD, anuleaza din vechiul program, inaintea inceperii transferului doar acele linii si variabile cu numere, respectiv nume deja existente in programul ce urmeaza a fi incarcat. Daca instructiunile VERIFY, LOAD si MERGE sint urmate de un sir vid ca nume al fisierului cautat, calculatorul va lucra asupra primului program pe care i-l intilneste.

Este posibil sa se inregistreze un program pe caseta, astfel incit atunci cind este reincarcat in memorie, el se lanseaza automat de la o linie specificata. Instructiunea este:

SAVE sir LINE numar

si face ca programul incarcat cu LOAD (dar nu si cu MERGE) sa

fie rulat automat de la linia specificata cu "numar". Daca nu este loc suficient in memorie, programul vechi si vechile variabile nu sint sterse si apare eroare:

Out of memory

In afara de programe si variabile se mai pot memora matrici si octeti. Pentru memorarea unei matrici se foloseste instructiunea:

SAVE sir DATA matrice()

unde

1. sir - este numele de pe banda al matricii
2. matrice - specifica numele matricii care va fi memorata (numerica sau sir de caractere).

Exemple

SAVE "test" DATA b()

In acest caz se salveaza pe caseta o matrice cu numele "test". Cind o gaseste trimite mejajul :

Number array: test

Matricea gasita este comparata cu matricea B din memorie.

LOAD "test" DATA b()

Se cauta matricea pe banda si daca este memorie libera suficienta, anuleazea o eventuala matrice B preexistenta, si incarca noua matrice pe banda denumind-o B.

MERGE nu poate fi folosit la inregistrarea matricilor pe banda.

Memorarea tip octet este folosita pentru orice tip de data, fara vreo referire asupra utilizarii acestei date. Memorarea tip octet se face cu:

SAVE sir CODE primul octet, numarul de octeti

Acest mod de memorare copiaza o parte din memoria interna a calculatorului, asa cum este, pe banda. Transferul in sens invers se face cu:

LOAD sir CODE adresa de inceput, lungime

Cind nu se specifica lungimea sirului de octeti, calculatorul va incarca toti octetii inregistrati pe caseta.

Exemplu

Zona de memorie in care se pastreaza imaginea pentru display incepe la adrasa 16384 si are 6912 octeti. Comanda:

SAVE "imagine" CODE 16384,6912

copiaza imaginea de pe ecran in momentul executiei comenzii, pe banda, cu numele imagine.

CODE 16384,6912 este folosita frecvent; de aceea a fost abreviata sub forma

SCREEN$

La memorarea imaginii video nu poate fi folosita comanda VERIFY.

Imprimanta

Cuprins: LLIST,LPRINT,COPY

Comenzile LPRINT si LLIST sint identice cu PRINT si

LIST, tiparind pe imprimanta, nu pe televizor.
Comanda COPY tipareste la imprimanta o copie a ecranului televizorului. COPY nu are efect in cazul listarilor automate (de cite ori se apasa CR).
Pentru a obtine un listing se poate folosi LIST urmat de COPY sau numai LLIST. Imprimanta poate fi oprita in timpul unei tipariri actionind BREAK.

Variabilele de sistem

Octetii din memorie de la adresa 23552 la adresa 23733 sint rezervati pentru operatii specifice ale sistemului. Ei pot fi cititi pentru a afla diferite lucruri despre sistem, iar citiva din ei pot fi si modificati. Acesti octeti se numesc variabile de sistem, si au cite un nume, dar nu trebuie confundati cu variabilele utilizate de BASIC. In cazul variabilelor formate din mai multi octeti, primul va fi octetul cel mai putin semnificativ.Variabilele de sistem sint date in lista de mai jos. Abrevierile din coloana 1 au urmatoarea semnificatie:

X aceasta variabila nu poate fi modificata deoarece sistemul va functiona eronat
N modificarea acestei variabile nu are un efect asupra functionarii normale a sistemului
n numarul de octeti din variabila

Tip	Adresa	Nume	Continut
N8	23552	KSTATE	Folosita in citirea tastaturii
N1	23560	LAST K	Retine ultima tasta apasata
1	23561	REPDEL	Durata (in 1/50 sec) cit trebuie tinuta apasata o tasta pentru a se repeta
1	23562	REPPER	Timpul (in 1/50 sec) dupa care se repeta o tasta apasata
N2	23563	DEFADD	Adresa argumentelor functiilor definite de utilizator
N1	23565	K DATA	Al doilea octet pentru controlul culorii introdus de la tastatura
N2	23566	TVDATA	Controlul culorii,al lui AT si TAB pentru TV
X38	23568	STRMS	Adresa canalului atasat caii
2	23606	CHARS	Adresa generatorului de caractere minus 256
1	23608	RASP	Durata sunetului de eroare
1	23609	PIP	Durata sunetului la apasarea unei taste
1	23610	ERR NR	Codul de mesaj minus 1
X1	23611	FLAGS	Diferiti indicatori de control ai sistemului BASIC
X1	23612	TVFLAG	Indicatori asociati cu televizorul
X2	23613	ERR SP	Adresa elementului din stiva masinii utilizat ca adresa de intoarcere in caz de eroare
N2	23615	LIST SP	Adresa de intoarcere la listarile automate
N1	23617	MODE	Specifica cursorul (K,L,C,E,G)
2	23618	NEWPPC	Linia la care se sare
1	23620	NSPPC	Numarul instructiunii in linie la care se sare
2	23621	PPC	Numarul liniei pentru instructiunea in executie
1	23623	SUBPPC	Numarul instructiunii din linie in executie
1	23624	BORDCR	Culoarea border-ului
2	23625	E PPC	Numarul liniei curente
X2	23627	VARS	Adresa variabilelor
N2	23629	DEST	Adresa variabilelor asignate
X2	23631	CHANS	Adresa datelor de canal
X2	23633	CURCHL	Adresa informatiei curente folosita pentru intrare sau iesire
X2	23635	PROG	Adresa programului BASIC
X2	23637	NXTLIN	Adresa urmatoarei linii din program
X2	23639	DATADD	Adresa ultimului element din lista DATA
X2	23641	E LINE	Adresa comenzii introduse
2	23643	K CUR	Adresa cursorului
X2	23645	CH ADD	Adresa urmatorului caracter care urmeaza sa fie interpretat
2	23647	XPTR	Adresa caracterului dupa semnul intrebarii
X2	23649	WORKSP	Adresa spatiului de lucru temporar
X2	23651	STKBOT	Adresa inferioara a stivei calculator
X2	23653	STKEND	Adresa de inceput a spatiului liber
N1	23655	BREG	Registru B al calculatorului
N2	23656	MEM	Adresa spatiului folosit pentru memoria calculatorului
1	23658	FLAGS2	Alti indicatori
X1	23659	DF SZ	Numarul liniilor din partea de jos a ecranului
2	23660	S TOP	Numarul liniei de sus a programului la listarea automata
2	23662	OLDPPC	Numarul liniei la care sare CONTINUE
1	23664	OSPCC	Numarul din linie la care sare CONTINUE
N1	23665	FLAGX	Diversi indicatori
N2	23666	STRLEN	Lungimea asignata sirului
N2	23668	T ADDR	Adresa urmatorului element din tabela sintaxa
2	23670	SEED	Variabila pentru RND
3	23672	FRAMES	Contorul de cadre
2	23675	UDG	Adresa primului grafic definit de utilizator
1	23677	COORDS	Coordonata x a ultimului punct plot-at
1	23678		Coordonata y a ultimului punct plot-at
1	23679	P POSN	Numarul pozitiei de scriere pe ecran
1	23680	PR CC	Octetul mai putin semnificativ al adresei pentru noua pozitie la care se imprima prin LPRINT
1	23681		Nefolosit
2	23682	ECHO E	Numarul coloanei si al liniei
2	23684	DF CC	Adresa de afisare pe ecran prin PRINT

2	23686	DFCCL	Acelasi lucru pentru partea de jos a ecranului
X1	23688	S POSN	Numarul coloanei pentru PRINT
X1	23689		Numarul liniei pentru PRINT
X2	23690	SPOSNL	Ca S POSN pentru partea de jos a ecranului
1	23692	SCR CT	Numara defilarile de ecran
1	23693	ATTR P	Culoarea curenta
1	23694	MASK P	Folosit pentru culori transparente
N1	23695	ATTR T	Culori temporare
N1	23696	MASK T	Ca MASK P dar temporar
1	23697	PFLAG	Alti indicatori
N30	23696	MEMBOT	Arie memorie calculator
2	23728		Nefolosit
2	23730	RAMTOP	Adresa ultimului din aria sistemului BASIC
2	23732	P-RAMT	Adresa ultimului octet de RAM

Canale I/O si cai

Cuprins: INPUT#,PRINT#,OPEN#,CLOSE#,LIST#,INKEY$#

Pentru fiecare echipament periferic sau port I/O este asignata o linie de comunicatie numita canal. Fiecarui canal existent i se poate asocia o parte componenta software numita cale. Pentru a transmite informatii pe un canal oarecare este suficient sa transmitem informatiile pe calea asignata acestui canal.

Exemplu

INPUT# s;'lista variabile'

citeste date de la portul asignat caii s si le asociaza variabilelor din lista de variabile. Similar

PRINT# s;'lista variabile'

trimite date catre portul asociat caii s.

Asignarea unei cai la un echipament I/O se face cu instructiunea OPEN# s,c unde:

s este numarul caii

c este un sir care specifica canalul

Instructiunea OPEN# realizeaza si initializarea echipamentului I/O. Unui canal i se pot asocia prin mai multe cai.

In configuratia de baza calculatorul HC - 85 recunoaste trei canale:

canalul K - claviatura

canalul S - ecran

canalul P - imprimanta

Canalele S si P sint canale pe care se poate doar scrie la echipamentul I/O.

Exemplu

```
10 OPEN# 5,"K"
20 PRINT# 5,"HC 90"
30 GO TO 20
```


trimite date la iesirea caii 5 care este asociata prin instructiunea OPEN# partii de jos a ecranului.

Pentru a anula asignarea caii s la un canal se foloseste instructiunea CLOSE# s. Dupa instructiunea CLOSE# calea s poate fi asociata altui canal.

La initializarea sistemului se deschid automat caile 0-3 , urmatoarea asignare :

calea 0 - canalul K

calea 1 - canalul K

calea 2 - canalul S

calea 3 - canalul P

Instructiunea LIST# s,n listeaza programul incepind cu linia n pe calea s.

Comanda INKEY$# s citeste un octet de pe calea s.

3.24 Alte echipamente

Retea

Poate fi folosita o periferie de tip retea pentru conectarea mai multor calculatoare HC-90 intre ele.

Interfata seriala

Interfata standard RS-232 permite conectarea unui HC-90 cu alt calculator sau alte periferice inzestrate cu aceasta interfata. Utilizarea se realizeaza folosind cuvintele cheie OPEN#, CLOSE#, MOVE, ERASE, CAT si FORMAT.

INTREPRINDEREA de CALCULATOARE ELECTRONICE

CERTIFICAT de GARANȚIE Nr.....

Produsul : FELIX HC 90

Seria:.....
Data fabricației: .....

Produsul este garantat pe o perioadă de 6 luni de la instalare și maxim 12 luni de la data vînzării. Acest termen se prelungește cu perioada de timp în care produsul a stat în reparație. Orice defecțiune care se datoarează unei utilizări necorespunzătoare sau manipulări defectuase sau neconformă cu instrucțiunile de instalare anulează garanția. Nu se consideră în garanție produsele care au fost desigilate de beneficiar. Cumpărătorul are obligația de a prezenta certificatul de garanție la întocmirea reclamației privitoare la defectarea produsului.
Produsele care sînt reparate în termenul de garanție de către intreprinderi neautorizate prin "agreement" de Intreprinderea de Calculatoare Electronice, își pierd garanția.

Șef Serviciu CTC

S-au făcut probe de funcționare, am primit manualul de instalare, manualul BASIC, caseta de demonstrații și cetificatul de garanție.

Semnătura beneficiar.....

Data vînzării.....

Semnătura și ștampila unității de desfacere